CONSTRUCTIONS OF GORENSTEIN RINGS

GUNTRAM HAINKE AND ALMAR KAID

1. INTRODUCTION

This is a summary of the fourth tutorial handed out at the CoCoA summer school 2005. We discuss two well known methods of construction of Gorenstein rings using Liaison Theory and so called Buchsbaum-Rim modules. In particular, we provide the necessary CoCoA (version 4.6) codes for implementation.

Let K be a field and $P := K[x_0, \ldots, x_n]$ the polynomial ring equipped with the standard grading.

Definition 1.1. Let $I \subset P$ be a homogeneous ideal. The graded *P*-algebra A := P/I is called *Gorenstein* if the following holds:

- (1) A is Cohen-Macaulay.
- (2) The last module in the minimal graded free resolution of A is a free module of rank one.

The K-vector space soc $A := ((0) :_A \mathfrak{m}) = \{x \in A : \mathfrak{m} \cdot x = 0\}$ is called the *socle* of A, where $\mathfrak{m} := (x_0, \ldots, x_n)$.

Proposition 1.2. Suppose A is Artinian. Then the following conditions are equivalent:

- (1) A is Gorenstein.
- (2) soc A is a K-vector space of dimension one.

Proof. Since A is Artinian we have dim A = 0. Hence A is Cohen-Macaulay. Note that soc $A \cong \operatorname{Hom}_P(K, A) = \operatorname{Ext}_P^0(K, A)$ and by [1, Exercise 3.3.26], we have dim_K $\operatorname{Ext}_P^0(K, A) = \dim_K \operatorname{Tor}_{n+1}^P(K, A)$. Since A is Artinian the Auslander-Buchsbaum formula (cf. [1, Theorem 1.3.3]) yields that dim_K $\operatorname{Tor}_{n+1}^P(K, A)$ equals the rank of the last free module in the graded free resolution of A and we are done.

Now we want to develop a CoCoA function which checks whether the algebra A = P/I is Gorenstein. To do this we use the following two subfunctions:

```
Define ProjDim(I)
   B:=BettiMatrix(I);
Return(Len(B[1]));
EndDefine;
Define IsCohenMacaulay(I)
   Pd:=ProjDim(I);
   Codim:=NumIndets()-Dim(CurrentRing()/I);
Return Pd=Codim;
EndDefine;
```

The function $\operatorname{ProjDim}(I)$ computes the projective dimension $\operatorname{pd} A$ of A = P/I. Next the function $\operatorname{IsCohenMacaulay}(I)$ checks whether Ais Cohen-Macaulay and returns the corresponding Boolean value. Note here that A is Cohen-Macaulay if and only if $\operatorname{pd} A = \operatorname{codim} I$, where $\operatorname{codim} I = \min{\{\operatorname{ht}(\mathfrak{p}) : I \subseteq \mathfrak{p} \text{ minimal}\}}.$

By using the functions ProjDim(I) and IsCohenMacaulay(I) we can define the function IsGorenstein(I) which checks whether A = P/I is Gorenstein and returns the corresponding Boolean value:

```
Define IsGorenstein(I)
   B:=BettiMatrix(I);
   C:=Len(B);
Return IsCohenMacaulay(I) AND (B[C][1])=1;
EndDefine;
```

2. Using Liaison Theory

In this section we use Liaison Theory to construct graded Gorenstein algebras. Our main tool is the following theorem (cf. also [3, Example 6.5(3)]).

Theorem 2.1. Let P/I_1 and P/I_2 be two graded Cohen-Macaulay rings of the same dimension d and suppose that $P/(I_1 \cap I_2)$ is Gorenstein. Moreover, suppose that I_1 and I_2 have no prime component in common. Then $P/(I_1 + I_2)$ is a graded Gorenstein algebra of dimension d - 1.

Proof. Let c := n + 1 - d denote the codimension of I_1 and I_2 . The minimal free resolutions of $I_1 \cap I_2$, I_1 and I_2 combined with the exact sequence

 $0 \longrightarrow I_1 \cap I_2 \longrightarrow I_1 \oplus I_2 \longrightarrow I_1 + I_2 \longrightarrow 0$

 $\mathbf{2}$

lead to the commutative diagram

If we apply the mapping cone to the resolutions of $I_1 \cap I_2$ and $I_1 \oplus I_2$ we get the (not necessarily minimal) free resolution

$$0 \longrightarrow P(-k) \longrightarrow \mathbb{F}_{c-1} \oplus (\mathbb{G}_c \oplus \mathbb{H}_c) \longrightarrow \cdots \longrightarrow \mathbb{F}_1 \oplus (\mathbb{G}_2 \oplus \mathbb{H}_2)$$
$$\longrightarrow \mathbb{G}_1 \oplus \mathbb{H}_1 \longrightarrow I_1 + I_2 \longrightarrow 0$$

for the ideal $I_1 + I_2$. Hence $\operatorname{pd} P/(I_1 + I_2) \leq c + 1$. Since I_1 and I_2 have no common component we have $\operatorname{codim} P/(I_1 + I_2) \geq c + 1$ and $\operatorname{dim} P/(I_1 + I_2) \leq n + 1 - (c + 1) = n - c$. In particular, depth $P/(I_1 + I_2) \leq \dim P/(I_1 + I_2) \leq n - c$ (cf. [1, Proposition 1.2.12]). The Auslander-Buchsbaum formula (cf. [1, Theorem 1.3.3]) gives $\operatorname{pd} P/(I_1+I_2) \geq n+1-(n-c) = c+1$. Hence $\operatorname{pd} P/(I_1+I_2) = c+1$. Again by the Auslander-Buchsbaum formula we get depth $P/(I_1+I_2) = n - c = \dim P/(I_1+I_2)$, i.e. $P/(I_1+I_2)$ is Cohen-Macaulay. Since $P/(I_1+I_2)$ has type one, it is also Gorenstein.

Definition 2.2. Let $V \subset \mathbb{P}^n$ be a subscheme with homogeneous coordinate ring $R := P/I_V$ (we assume that I_V is saturated).

- (1) V is arithmetically Cohen-Macaulay if R is Cohen-Macaulay.
- (2) V is arithmetically Gorenstein if R is Gorenstein.

Theorem 2.1 has the following geometric formulation: Let $V_1, V_2 \subset \mathbb{P}^n$ be arithmetically Cohen-Macaulay subschemes of codimension c with no common component. Assume that $X := V_1 \cup V_2$ is arithmetically Gorenstein. Then $V_1 \cap V_2$ is arithmetically Gorenstein of codimension c + 1.

For instance, consider two arithmetically Cohen-Macaulay curves which are geometrically linked, i.e. they have no common component and their union is a complete intersection. In this case Theorem 2.1 asserts that the intersection of these curves is an arithmetically Gorenstein set of points. To illustrate this we provide the following example.

Example 2.3. We start with the twisted cubic curve C in \mathbb{P}^3 , i.e. the curve defined by the ideal I_C which is generated by the 2×2 minors of the matrix

$$\begin{pmatrix} x_0 & x_1 & x_2 \\ x_1 & x_2 & x_3 \end{pmatrix}.$$

To realise this situation in CoCoA we use the command

Use P::=Q[x[0..3]];

to work in the polynomial ring in four variables. We get the ideal I_C in the following way:

Matrix:=Mat([[x[0],x[1],x[2]],[x[1],x[2],x[3]]]);

```
IC:=Ideal(Minors(2,Matrix));
```

By using

```
IsCohenMacaulay(IC);
```

we see that C is an arithmetically Cohen-Macaulay curve. Furthermore we use the following function:

```
Define IsCompleteIntersection(I)
  Codim:=NumIndets()-Dim(CurrentRing()/I);
Return Len(MinGens(I))=Codim;
EndDefine;
```

It checks whether P/I is a complete intersection in terms of the length of a minimal system of generators.

If we apply IsCompleteIntersection(I) to the ideal I_C , we see that the twisted cubic curve is not a complete intersection. Now we write another very useful function:

```
Define GenRegSeq(I,L)
  M:=Ideal(Indets());
  Seq:=[];
  N:=Ideal(0);
```

```
Foreach T In L Do
    J:=Intersection(M<sup>T</sup>,I);
    D:=Deg(Head(MinGens(J)));
    If D<>T Then
      PrintLn"No regular sequence of this type possible";
      Return;
    EndIf;
    L2:=[F In MinGens(J) | Deg(F)=D];
    S:=Sum([Rand(-10,10)*F | F In L2]);
    If N:Ideal(S)<>N Then
      PrintLn"No regular sequence of this type possible";
      Return;
    EndIf;
    Append(Seq,S);
    N:=Ideal(Seq);
  EndForeach;
Return Seq;
EndDefine;
```

The function GenRegSeq(I,L) takes an ideal I and a list of degrees L and computes a regular sequence of homogeneous polynomials in I whose degrees are given by L. If this is not possible it conveys this information.

Now we get by

ID:=Ideal(GenRegSeq(IC,[3,3]));

the ideal I_D of a complete intersection curve D of type (3,3) containing C. Moreover, we get the vanishing ideal I_E of the residual curve $E := D \setminus C$ by computing the colon ideal

IE:=ID:IC;

We use the command

Hilbert(P/IE);

to see that the Hilbert polynomial of E is $\operatorname{HP}_E(t) = \operatorname{HP}_{P/I_E}(t) = 6t-2$ for $t \geq 1$. Hence E is a (arithmetic) genus three curve of degree 6. Further, with the help of

IsCohenMacaulay(IE);

we notice that E is arithmetically Cohen-Macaulay.

Now set $\mathbb{X} := C \cap E$. By Theorem 2.1 the scheme \mathbb{X} is an arithmetically Gorenstein set of points. We compute its vanishing ideal $I_{\mathbb{X}}$ by

IX:=IE+IC;

The command

Hilbert(P/IX);

yields the constant Hilbert polynomial $\operatorname{HP}_{\mathbb{X}}(t) = \operatorname{HP}_{P/I_{\mathbb{X}}}(t) = 8$ for $t \geq 3$ of X. Hence X is a zero-dimensional scheme of degree 8. In particular, if we use our two CoCoA functions IsGorenstein(I) and IsCompleteIntersection(I) applied to the ideal $I_{\mathbb{X}}$, we verify that X is arithmetically Gorenstein but not a complete intersection. Thus we have constructed a "new" 1-dimensional graded Gorenstein ring $P/I_{\mathbb{X}}$.

Example 2.4. We will give an analog example to the previous one and construct the ideal X of 11 points on the twisted cubic curve C defined above. Here, we utililize a complete intersection of type (3, 4). So we get in the same way the ideal I_X via the commands:

ID:=Ideal(GenRegSeq(IC,[3,4]));

IE:=ID:IC; IX:=IE+IC;

To check whether X is a reduced scheme, we query

IX=Radical(IX);

and get $I_{\mathbb{X}} = \operatorname{rad}(I_{\mathbb{X}})$. If we compute the Hilbert polynomial of \mathbb{X} with Hilbert (P/IX);

we get indeed $\operatorname{HP}_{\mathbb{X}}(t) = \operatorname{HP}_{P/I_{\mathbb{X}}}(t) = 11$ for $t \geq 4$. Hence (at least after extension of the base field) \mathbb{X} is a set of 11 points in \mathbb{P}^n . We verify Theorem 2.1 with

IsGorenstein(IX);
IsCompleteIntersection(IX);

and notice that X is arithmetically Gorenstein but not a complete intersection.

3. Using Buchsbaum-Rim modules

We still work in the polynomial ring $P = K[x_0, \ldots, x_n]$ over a field K. First we recall the notion of a *Buchsbaum-Rim module*.

Definition 3.1. Let $F := \bigoplus_{i=1}^{t+r} P(-a_i)$ and $G := \bigoplus_{j=1}^{t} P(-b_j)$ be two graded free *P*-modules, where $t, r \in \mathbb{N}$ and $a_1, \ldots, a_{t+r}, b_1, \ldots, b_t \in \mathbb{Z}$. Furthermore, let $\Phi : F \to G$ be a homogeneous map. The map Φ is given by a $t \times (t+r)$ matrix \mathcal{A} whose entries are homogeneous polynomials. The module $B := \ker \Phi$ is called the *Buchsbaum-Rim*

6

module associated to Φ (or to \mathcal{A}) if the ideal of maximal minors of \mathcal{A} has the expected codimension r + 1.

Remark 3.2. The sheaf \tilde{B} on \mathbb{P}^n associated to B is also called the *Buchsbaum-Rim sheaf* associated to Φ (or to \mathcal{A}). In the following it will be not necessary to use the sheaf framework.

We recall that the *top dimensional part* I^{top} of an ideal I is the intersection of its primary components of maximal dimension. The following theorem (cf. [4, Theorem 4.7]) provides a method of constructing Gorenstein algebras via Buchsbaum-Rim modules.

Theorem 3.3. Let B be the Buchsbaum-Rim module associated to a map $\Phi : \bigoplus_{i=1}^{t+r} P(-a_i) \to \bigoplus_{j=1}^{t} P(-b_j)$, $s \in B$ a generic homogeneous element and let $I \subseteq P = K[x_0, \ldots, x_n]$ be the ideal generated by the components of s. If r < n is odd, then the top dimensional part I^{top} yields a Gorenstein ring P/I^{top} .

Now we give an example of an application of Theorem 2.1.

Example 3.4. Let $C \subset \mathbb{P}^4$ be the curve defined by the vanishing of the 2×2 minors of the matrix

$$\begin{pmatrix} x_0 & x_1 & x_2 & x_3 \\ x_1 & x_2 & x_3 & x_4 \end{pmatrix}.$$

In the following we apply Theorem 3.3 to find an arithmetically Gorenstein curve D of small degree containing it. We use the command

Use P::=Q[x[0..4]];

to work with CoCoA in the polynomial ring in five variables over the rationals. Now we want to compute the Buchsbaum-Rim module B associated to the matrix $\mathcal{A} := (x_0, x_1, x_2, x_3)$. To do this, we compute the syzygy module for the variables x_0, x_1, x_2, x_3 :

B:=Syz(x[0]..x[3]);

We notice that B is contained in the free P-module $F := P(-1)^4$. Further, we get the vanishing ideal I_C of the curve C by

IC:=Ideal(Minors(2,Mat([x[0]..x[3],x[1]..x[4]])));

Since we want D to contain C, we use an element $s \in B_d$, $d \in \mathbb{Z}$, whose components are contained in I_C . Thus we have to compute $B' := B \cap I_C F$. Firstly, we compute the module $I_C F$ in the following way:

```
GenIC:= Gens(IC);
ICF:=[];
For I:=1 To 4 Do
Foreach X In GenIC Do
Append(ICF,X * E_(I,4));
EndForeach;
EndFor;
ICF:=Module(ICF);
```

Hence we get B' by

Bprime:=Intersection(B,ICF);

Now we construct a generic element $s \in B'$ of degree 2. To do this, we firstly compute all generators of degree 2 of B' by

Bprime_2:=[X In Gens(Bprime)|Deg(X)=2];

To get a generic (in the randomized sense) element $s \in B'_2$ we use the command Rand(-10,10) which produces a random integer between -10 and 10 (of course one can use other parameter or use Rand() to produce an "arbitrary random integer"):

S:=Sum([Rand(-10,10)*X | X In Bprime_2]);

We denote by I the ideal generated by the components of s and realize this with

I:=Ideal([X | X In List(S)]);

Let D be the projective scheme defined by $I(=I_D)$. To describe D, we use the function EquiIsoDec(I) which computes a list of unmixed ideals I_1, \ldots, I_k such that $rad(I) = rad(I_1) \cap \cdots \cap rad(I_k)$. So the computation

L:=EquiIsoDec(I);

yields ideals L[2], L[3], L[5], L[6] (L[1] and L[4] equal the unit ideal and are therefore obviously redundant). We use the function Radical(I) to check that all these ideals are radical ideals. Next, we use the function Intersection(I,J) to conclude the inclusions $L[2] \subset L[3], L[2] \subset L[5]$ and $L[2] \subset L[6]$. Hence rad(I) = L[2]. Next we check by

Hilbert(P/I);

Hilbert(P/L[2]);

that P/I has Hilbert polynomial $\operatorname{HP}_{P/I}(t) = 5t + 1$ and P/L[2] has Hilbert polynomial $\operatorname{HP}_{P/L[2]}(t) = 5t$. Therefore they differ by a "point" in the sense of the exact sequence

$$0 \longrightarrow L[2]/I \longrightarrow P/I \longrightarrow P/L[2] \longrightarrow 0.$$

```
8
```

Finally, we want to compute the top dimensional part of D, i.e. I_D^{top} . The following lemma provides an effective method of computing the top dimensional part of an ideal.

Lemma 3.5. Let $I \subseteq P$ be a saturated ideal and $J \subseteq I$ be an ideal generated by a regular sequence whose length is the codimension of I. Then

 $J:_P (J:_P I) = I^{\operatorname{top}}.$

Proof. For a detailed proof see [2, Proposition 5.2.3(d), Remark 5.1.5 and the proof on p. 123]. If X denotes the arithmetically Gorenstein scheme defined by J, then basically one shows that the projective schemes defined by the ideals I^{top} and $J :_P I$ are algebraically CI-linked via X.

Now we apply Lemma 3.5 to our example. So we compute an ideal J generated by a regular sequence of length codim I = 3 in I:

J:=Ideal(GenRegSeq(I,[2,2,2]));

According to Lemma 3.5 we get the top dimensional part of I by

Itop:=J:(J:I);

To describe $D^{\text{top}} := \operatorname{Proj}(P/I^{\text{top}})$ we use

Hilbert(P/Itop);

and get $\operatorname{HP}_{D^{\operatorname{top}}}(t) = \operatorname{HP}_{P/I^{\operatorname{top}}}(t) = 5t$ for $t \ge 1$, i.e. D^{top} has arithmetic genus $p_a(D^{\operatorname{top}}) = 1$ and degree 5. By Theorem 3.3 the curve D^{top} is arithmetically Gorenstein. We can describe D^{top} a bit more. We check via

Intersection(IC,Itop)=Itop;

that $I_{D^{\text{top}}} \subset I_C$, i.e. the curve C is contained in D^{top} . Further, we compute the residual curve $E := D^{\text{top}} \setminus C$ with

IE:=Itop:IC;

and its Hilbert polynomial by

Hilbert(P/IE);

We get $\operatorname{HP}_E(t) = \operatorname{HP}_{P/I_E}(t) = t + 1$ for $t \ge 0$. Hence the top dimensional part D^{top} of D is the union of C with a line.

References

 W. Bruns and J. Herzog. Cohen-Macaulay Rings, Revised edition. Cambridge University Press, 1998.

- [2] J. Migliore. Introduction to Liaison Theory and Deficiency Modules., volume 165. Birkhäuser, Progress in Mathematics, 1998.
- [3] J. Migliore. Experiments in commutative algebra and algebraic geometry fourth international CoCoA school on computer algebra, Porto Conte, May 23-26. In this volume, 2005.
- [4] J. Migliore, U. Nagel, and C. Peterson. Buchsbaum-Rim sheaves and their multiple sections. J. Algebra, 219:378–420, 1999.

Fakultät für Mathematik, Universität Bielefeld, D-33501 Bielefeld, Germany

 $E\text{-}mail\ address:\ \texttt{ghainkeQmath.uni-bielefeld.de}$

Department of Pure Mathematics, University of Sheffield, Hounsfield Road, Sheffield S3 7RH, United Kingdom

E-mail address: A.Kaid@sheffield.ac.uk